

Lesson 20

Parametric Equations

Initializations

```
> restart;  
with(plots):
```

20.1 Parametric Equations

In stead of using a Cartesian equation, curves can be expressed in terms of parametric equations of the form

$$x = f(t) \quad y = g(t) \quad a \leq t \leq b$$

Mathematical details will be provided in class.

Examples

Example 20.1.1

Plot the curve with parametric equations

$$x = -2t + 4 \quad y = t^2 - 3t + 7 \quad -3 \leq t \leq 3$$

and compute its Cartesian equation.

Solution

Maple syntax for plotting parametric curves is given by

plot([x(t), y(t), t = a..b]);

Since the variable names x and y are used so frequently, it is not a good idea to associate them with any particular function. Instead, we let

$$f(t) = x(t) = -2t + 4 \quad \text{and} \quad g(t) = y(t) = t^2 - 3t + 7$$

```
> f:=t->-3*t+4;
```


```
g:=t->t^2-3*t+7;
```

```
f:=t->-3*t+4
```

```
g:=t->t^2-3*t+7
```

(2.1.1.1)

```
> plot([f(t), g(t), t=-3..3]);
```



The Cartesian equation of this curve can be found eliminating t from the equations

$$x = f(t) \quad \text{and} \quad y = g(t)$$

```
> eq1:=x=f(t);
  eq2:=y=g(t);
```

$$eq1 := x = -3t + 4$$

$$eq2 := y = t^2 - 3t + 7 \quad (2.1.1.2)$$

Solve the first equation for t and substitute the result into the second equation.

```
> tv:=isolate(eq1, t);
```

$$tv := t = -\frac{1}{3}x + \frac{4}{3} \quad (2.1.1.3)$$

```
> cartesian:=simplify(subs(tv, eq2));
```

$$cartesian := y = \frac{1}{9}x^2 + \frac{1}{9}x + \frac{43}{9} \quad (2.1.1.4)$$

This equation represents a parabola, which is consistent with the graphical image above.

Example 20.1.2

Consider $x = f(t) = 2 \sin t$ and $y = g(t) = 5 \cos t$. Plot the curve and find its Cartesian equation.

Solution


Because both $f(t)$ and $g(t)$ are periodic modulo 2π , we plot the curve for $0 \leq t \leq 2\pi$.

```
> f:=t->2*sin(t);
  g:=t->5*cos(t);
```

$$f := t \rightarrow 2 \sin(t)$$

$$g := t \rightarrow 5 \cos(t) \quad (2.1.2.1)$$

```
> plot([f(t), g(t), t=0..2*Pi], scaling=constrained);
```



The Cartesian equation can again be found by eliminating the variable t from the parametric equations.

```
> eq1:=x=f(t);
    eq2:=y=g(t);
```

$$eq1 := x = 2 \sin(t)$$

$$eq2 := y = 5 \cos(t) \quad (2.1.2.2)$$

```
> tv:=isolate(eq1, t);
```

$$tv := t = \arcsin\left(\frac{1}{2} x\right) \quad (2.1.2.3)$$

```
> c1:=subs(tv, eq2);
```

$$c1 := y = 5 \cos\left(\arcsin\left(\frac{1}{2} x\right)\right) \quad (2.1.2.4)$$

Simplify the result by expanding the trigonometric expressions.

```
> c2:=expand(c1);
```

$$c2 := y = \frac{5}{2} \sqrt{4 - x^2} \quad (2.1.2.5)$$

Square both sides by mapping the function $u \rightarrow u^2$ onto this equation.

```
> c3:=map(u->u^2, c2);
```

$$c3 := y^2 = 25 - \frac{25}{4} x^2 \quad (2.1.2.6)$$

It is an ellipse.

▼ Example 20.1.3

One of the major advantages of a parametric representation is that it may be interpreted as the position at time t of a particle tracing the curve. We illustrate this by considering the path defined by

$$x(t) = t + 2 \sin 2t \quad \text{and} \quad y(t) = t + 2 \cos 5t$$

- i) Plot the curve for $-5 \leq t \leq 5$.
 ii) Animate the movement of a particle tracing the curve for these values of t .

Solution

- i) Plot the curve for $-5 \leq t \leq 5$.

Code the parametric equations and plot.


```
> f:=t->t+2*sin(2*t);
   g:=t->t+2*cos(5*t);
```

$$f := t \rightarrow t + 2 \sin(2t)$$

$$g := t \rightarrow t + 2 \cos(5t)$$

(2.1.3.1)

```
> plot([f(t), g(t), t=-5..5]);
```



- ii) Animate the movement of a particle tracing the curve for these values of t .

We create 100 images showing the particle, visualized as a blue box, and the part of the curve traced by the particle at time t . Then we feed the images into the digital equivalent of a VCR, and run the animation/

```
> for k to 100 do
   p||k:=plot([f(t), g(t), t=-5..-5+10/100*k]):
   q||k:=pointplot([f(-5+10/100*k), g(-5+10/100*k)], symbol=
   box, color=blue):
   od:
> r:=[seq(display([p||k, q||k]), k=1..100)]:
> display(r, insequence=true);
```

